

Combinatorics of Non-Crossing Configurations

F. Cazals, August 1997

Generalities

Non-Crossing configurations

Take n points equally spaced on the unit circle, and draw chords between these points with the constraint that no two chords cross one-another. The resulting configuration is called a **non-crossing configuration** and the study of such entities originates in the work of Euler and Segner in 1753 for counting triangulations of a n -gon. Since then several types of such configurations have been defined, and for example the presence/absence of cycles and the number of connected components in a given configuration define the classes of trees and forests, connected graphs and general graphs. In addition to be of combinatorial interest per se, these configurations are also important for algorithmic problems arising in computer graphics or computational geometry where they provide simple models for real-world situations.

If historically the study of these configurations has been carried out one at a time, it turns out that they all fit in the model of algebraic and analytic combinatorics and are thus amenable to a unified treatment. More precisely, they can be defined in terms of grammars, from which the generating functions can automatically be obtained and used to asymptotically analyse the number of configurations, the number of connected components, etc.

The goal of this worksheet is to present this chain, from the grammars specification using `comstruct`, to the asymptotics of algebraic functions using some features of `gfun`, together with some plots of random configurations and of the singularities determining the asymptotic behaviour. The reader is referred to [FlaNo97] for the details.

The worksheet is organized as follows. In the rest of this section we define a few functions of null combinatorial interest but that we shall need to plot random configurations. In section two we present the grammar specifications for 6 of the main non-crossing configurations together with some plots of random configurations. And section three is devoted to the asymptotic machinery used to count the number of configurations of a given size, as well as a comparison between the asymptotic estimates and the exact values.

But to begin with, we first load the `comstruct`, `gfun` and `plots` libraries:

```
> with(comstruct):with(gfun):with(plots):
```

Appendix

Again, we define here a few functions we shall need to plot NC configurations. The first one returns the number of atoms in a structure, that is its size:

```
> size:=proc(t) convert(map(size, t), `+`) end:size(Epsilon):=0: size(Z):=1:
```

Since a configuration is defined by edges drawn between points equally spaced on the unit circle, we first show how to retrieve these pairs of indices from the grammars to be defined in the next section:

Collecting the edges of a NC-tree

Since a tree T satisfies $T = \text{Product}(Z, \text{Sequence}(\text{Butterflies}))$, to plot it we just have to collect its edges which are pairs of indices in $0..n-1$:

```
> plotTree:= proc(aTree) local r, nbVertices;
```

```

#--op returns the Sequence; 0 is the index of the root on the circle;
#--1 indicates that the butterflies attached to the root are counted as a
#--right wing
r:=getTreeEdges(op(2, aTree), 0, 1);
nbVertices:=size(aTree);

```

```

plotSetOfEdges(r, nbVertices)
end;

```

The parameters of `getTreeEdges` are the following: `aSeqOfBtf` is a sequence of butterflies; `rootIdx` is the index of the vertex this sequence is attached to; `leftRight` (-1 or +1) indicates if this sequence is a left/right wing of the bug. The algorithm works as follows:

1. We first compute the indices of the apex vertices of the butterflies
2. For a given butterfly, we recurse on the 2 wings and attach its apex to the `rootIdx` parameter

```

> getTreeEdges:=proc(aSeqOfBtf, rootIdx, leftRight)
local i, a, eL, currentBtf, cumul;

if aSeqOfBtf = Epsilon then {}
else
cumul := rootIdx;
if leftRight=1 then #--we are processing a right wing
for i to nops(aSeqOfBtf) do
currentBtf := op(i, aSeqOfBtf);
a[i] := cumul + size(op(1, currentBtf)) + 1;
cumul := cumul + size(currentBtf)
od;
else #--and here a left one
for i from nops(aSeqOfBtf) by -1 to 1 do
currentBtf := op(i, aSeqOfBtf);
a[i] := cumul - size(op(3, currentBtf)) - 1;
cumul := cumul - size(currentBtf);
od;
fi;

#-- recurses for each Prod(?, Z, ?) in the sequence, with ? = E or Seq()
eL := {};
for i from 1 to nops(aSeqOfBtf) do
currentBtf := op(i, aSeqOfBtf);
eL := eL union getTreeEdges(op(1, currentBtf), a[i], -1);
eL := eL union {[a[i], rootIdx]};
eL := eL union getTreeEdges(op(3, currentBtf), a[i], 1);
od;

#--returns the result
eL;
fi;
end;

```

Collecting the edges of a NC-graph

This first procedure recursively collects the edges of an EA, and is a straightforward application of the EA definition above. The parameter `ori` stands for the index of the leftmost point of the arch:

```

> getArchEdges:=proc(arch, ori) local i, offset, res;

if (arch=Epsilon) then res:={}

```

```

else
#--we first add the `roof' of the arch
res := {[ori, ori+size(arch)]};

#-----Prod(Z, Seq1, Seq2)
if (op(1,arch)=Z) then
res := res union getArchEdges(op(2,arch), ori);
res := res union getArchEdges(op(3,arch), ori+1+size(op(2,arch)))
else #--Sequence of arches
offset:=0;
#--let's process all the arches in this sequence
for i from 1 to nops(arch) do
res := res union getArchEdges(op(i, arch), ori+offset);
offset := offset + size(op(i,arch))
od
fi
res;
end:

```

Same thing but to a sequence of arches:

```

> getArchesSeqEdges:=proc(archesSeq, ori)
local i, arch, offset, res;
res:={}; offset:=0;

for i from 1 to nops(archesSeq) do
arch:= op(i, archesSeq);
res:= res union getArchEdges(arch, ori+offset);
offset := offset+size(arch)
od;
#--returns the setOfEdges and the new origin
res,ori+offset;
end:

```

And now the main procedure which collects the edges of a Non-Crossing graph:

```

> getNCGraphEdges:=proc(aNCGraph)
local res, ori, edges,
i, j,
seqOfSeqOrProd, seqOrProd;

#--first, the right ear which is a Seq(EA)
ori:=1;
res:=getArchesSeqEdges(op(3,aNCGraph), ori);
edges:=res[1]; ori := res[2];

edges := edges union {[0, ori]};

#--then the EAs inbetween two successive childs of v_1
seqOfSeqOrProd := op(5,aNCGraph);

for i from 1 to nops(seqOfSeqOrProd) do
seqOrProd:=op(i, seqOfSeqOrProd);
#--2 connected graphs
if (op(1, seqOrProd)=Z) then
res:=getArchesSeqEdges(op(2, seqOrProd), ori);
edges:= edges union res[1]; ori := res[2];

```

```
res:=getArchesSeqEdges(op(3, seqOrProd), 1+ori);
edges:= edges union res[1]; ori := res[2]
```

```
!--a Sequence
```

```
else
res:=getArchesSeqEdges(seqOrProd, ori);
edges:= edges union res[1]; ori := res[2]
fi;
```

```
!--we need to add the current child to the graph root
edges := edges union {[0, ori]};
od;
```

```
!--and the left ear
```

```
res:=getArchesSeqEdges(op(4,aNCGraph), ori);
edges:= edges union res[1]; ori := res[2];
```

```
!--returns the result
```

```
edges
end:
```

To plot a NC Graph, we just collect the edges and pass them to the plotSetOfEdges procedure:

```
> plotNCGraph:=proc(aNCGraph);
plotSetOfEdges(getNCGraphEdges(aNCGraph), size(aNCGraph))
end:
```

Now, given pairs of indices in $0, \dots, \text{nbVertices}-1$ on the unit circle, the following procedure draws the corresponding chords

assuming that the k -th point has coordinates $((\cos((2 \text{ Pi } k)/\text{nbVertices}), \sin((2 \text{ Pi } k)/\text{nbVertices}))$):

```
> plotSetOfEdges:=proc(aSetOfEdges, nbVertices) local pointsOnCircle;
```

```
pointsOnCircle:= expand(map(`*`, aSetOfEdges, 2*Pi/nbVertices));
plot([op(map2(map, [cos,sin], pointsOnCircle))], color=blue, axes=NONE)
end:
```

Counting and drawing non-crossing configurations

Trees and forests

Trees

A Tree is a sequence of butterflies attached to a root, a Butterfly being an ordered pair of trees whose roots have been merged into a single node. Since this merge step cannot be specified by an operation such as $B = \text{Prod}(T, Z, T)/Z$ in combstruct, it is more convenient to express a butterfly as the product of 2 forests, a Forest being a sequence of butterflies. Indeed, we now just have to attach the roots of all the trees of the two forests to a newly added node:

```
> tbf:={T=Prod(Z, Sequence(B)), F=Sequence(B), B=Prod(F,Z,F)};
```

$$tbf := \{T = \text{Prod}(Z, \text{Sequence}(B)), F = \text{Sequence}(B), B = \text{Prod}(F, Z, F)\}$$

Standard functionalities of Combstruct consist in counting the number of entities of a given type:

```
> seq(count([T,tbf], size=i), i=1..10);
```

1, 1, 3, 12, 55, 273, 1428, 7752, 43263, 246675

> **seq(count([F,tbf], size=i),i=1..10);**

1, 3, 12, 55, 273, 1428, 7752, 43263, 246675, 1430715

> **seq(count([F,{F=Sequence(B), B=Prod(F,Z,F)}],size=i),i=0..10);**

1, 1, 3, 12, 55, 273, 1428, 7752, 43263, 246675, 1430715

And randomly generating all the configurations of a given size:

> **allstructs([T,tbf],size=2);**

[Prod(Z, Sequence(Prod(E, Z, E)))]

> **expleT:=draw([T, tbf], size=20);**

expleT := Prod(Z, Sequence(Prod(Sequence(Prod(E, Z, Sequence(Prod(Sequence(Prod(E, Z, Sequence(Prod(Sequence(Prod(Sequence(Prod(E, Z, Sequence(Prod(Sequence(Prod(E, Z, Sequence(Prod(E, Z, Sequence(Prod(E, Z, Sequence(Prod(E, Z, E))))), Z, E))), Z, E))), Z, E))), Z, E))), Z, E))), Z, E))), Z, E))), Z, E))), Z, E)))

We can also use gfseries to retrieve the first terms of the generating functions:

> **Order:=10:gfseries(tbf,unlabelled,z);**

table([

$$F(z) = 1 + z + 3z^2 + 12z^3 + 55z^4 + 273z^5 + 1428z^6 + 7752z^7 + 43263z^8 + 246675z^9 + O(z^{10})$$

$$B(z) = z + 2z^2 + 7z^3 + 30z^4 + 143z^5 + 728z^6 + 3876z^7 + 21318z^8 + 120175z^9 + O(z^{10})$$


$$T(z) = z + z^2 + 3z^3 + 12z^4 + 55z^5 + 273z^6 + 1428z^7 + 7752z^8 + 43263z^9 + O(z^{10})$$

$$Z(z) = z$$

])

And here is an example of random tree:

> **plotTree(draw([T,tbf],size=20));**



>

Forests

In order to dissociate the trees in a forest, let us substitute to each vertex of a tree another vertex together with a forest

```
> fo:={B=Prod(Sequence(B),V,Sequence(B)),
V=Union(Prod(Z, F)),
F=Union(Epsilon, Prod(V, Sequence(B)))};
```

```
fo := {B = Prod(Sequence(B), V, Sequence(B)), V = Union(Prod(Z, F)), F = Union(E, Prod(V, Sequence(B)))}
```

```
> seq(count([F, fo], size=i), i=1..10);
```

```
1, 2, 7, 33, 181, 1083, 6854, 45111, 305629, 2117283
```

Connected and general graphs

Connected graphs

To see how nc-graphs are built, consider the set of childs v_2, v_{i+1}, \dots, v_j attached to the root v_1 .

The graphs built on v_2, \dots, v_i and $v_j, \dots, v_n - 1$ are connected by hypothesis, while between any two other vertices v_k and v_{k+1} one can have either one connected graph or two connected graphs.

But any graph built from a sequence of successive vertices of the circle is a system of arches since the arcs which are chords are not allowed to cross. The endpoints of these arches are shared by the graphs built to the left and to the right of a given v_k , so that we shall say that the size of an arch built on n points is $n - 1$. At last, we are interested here in Elementary Arches, that is arches that always contain an arc between the first and last points. General arches are easily obtained by sequencing EAs.

From this discussion we derive the Combstruct specification of NC graphs containing at least 2 vertices. In particular, the 5 arguments of a C entity are as follows:

1.first Z: the root of the graph

2.second Z: the leftmost point of the first EA

3 and 4. the two EAs built on v_2, \dots, v_i and $v_j, \dots, v_n - 1$

5.the sequence of EAs found between two consecutive childs of v_1 . The first term corresponds to one EA,

and the second one to two EAs. For the latter case, the Z in the Prod stands for the leftmost point of the second EA.

```
> ar:={EA = Union(Sequence(EA, card >= 2),
Prod(Z, Sequence(EA), Sequence(EA))
),
C=Union(Z,
Prod(Z,Z,Sequence(EA), Sequence(EA),
Sequence(Union(Sequence(EA,card>=1), Prod(Z,Sequence(EA),Sequence(EA))))))};
```

```
ar := {EA = Union(Sequence(EA, 2 ≤ card), Prod(Z, Sequence(EA), Sequence(EA))), C = Union(Z, Prod(Z, Z,
Sequence(EA), Sequence(EA),
```

```
Sequence(Union(Sequence(EA, 1 ≤ card), Prod(Z, Sequence(EA), Sequence(EA))))))}
```

We can now count the number of elementary arches of a given size. The sequence found is not in [Sloa95]:

```
> seq(count([EA,ar], size=i),i=1..20);
```

```
1, 3, 16, 105, 768, 6006, 49152, 415701, 3604480, 31870410, 286261248, 2604681690, 23957864448, 222399744300,
2080911654912, 19604537460045, 185813170126848, 1770558814528770, 16951376923852800,
162984598242674670
```

```
> allstructs([EA,ar],size=2);
```

```
[ Sequence(Prod(Z, E, E), Prod(Z, E, E)), Prod(Z, Sequence(Prod(Z, E, E)), E), Prod(Z, E, Sequence(Prod(Z, E, E)))
]
```

We can also count the number of NC-graphs. The corresponding sequence turns out to be M3594 in [Sloa95] and gives the reverse

of the g.f. for squares:

```
> seq(count([C,ar], size=i),i=1..10);
```

```
1, 1, 4, 23, 156, 1162, 9192, 75819, 644908, 5616182
```

As usual, we can get all the structures of a given size, or just draw some of them:

```
> allstructs([C,ar],size=3);
```


```
[ Prod(Z, Z, E, E, Sequence(Sequence(Prod(Z, E, E)))), Prod(Z, Z, E, E, Sequence(Prod(Z, E, E))),
Prod(Z, Z, E, Sequence(Prod(Z, E, E)), E), Prod(Z, Z, Sequence(Prod(Z, E, E)), E, E)]
```

```
> draw([EA,ar],size=10);
```

```
Sequence(Sequence(Prod(Z, E, E),
Prod(Z, E, Sequence(Prod(Z, E, Sequence(Prod(Z, E, Sequence(Prod(Z, E, E))))))), Prod(Z, E, E)),
Sequence(Sequence(Prod(Z, E, E), Prod(Z, E, E), Prod(Z, E, E)), Prod(Z, E, E)))
```

And an example of random graph:

```
> plotNCGraph(draw([C,ar],size=10));
```



>

General graphs

As observed in [FlaNo97], a general graph is obtained from a connected one by the substitution $Z \rightarrow \text{Prod}(Z, G)$. So that we just have to rewrite the previous grammar by adding a new symbol which makes this substitution. Notice however that the decomposition of a connected graph misses a configuration for general graphs: the one where vertex v_1 does not have any child, which we therefore add:

```
> br:= {EA = Union(Sequence(EA, card >= 2),
Prod(V, Sequence(EA), Sequence(EA))
),
V=Union(Prod(Z, G)),
G=Union(Epsilon,
Prod(Z, G),
Prod(V, V, Sequence(EA), Sequence(EA),
Sequence(Union(Sequence(EA, card >= 1), Prod(V, Sequence(EA), Sequence(EA))))))
}
};
```

```
br := {EA = Union(Sequence(EA, 2 ≤ card), Prod(V, Sequence(EA), Sequence(EA))), V = Union(Prod(Z, G)), G =
Union(E, Prod(Z, G), Prod(V, V, Sequence(EA), Sequence(EA),
Sequence(Union(Sequence(EA, 1 ≤ card), Prod(V, Sequence(EA), Sequence(EA))))))}
```

The number of graphs is given by the following sequence, not to be found in [Sloa95]:

```
> ggSeq:= [seq(count([G, br], size=i), i=0..20)];
```

```
ggSeq := [1, 1, 2, 8, 48, 352, 2880, 25216, 231168, 2190848, 21292032, 211044352, 2125246464, 21681954816,
223623069696, 2327818174464, 24424842461184, 258054752698368, 2742964283768832, 29312424612462592,
314739971287154688]
```

In this case, it also turns out that the differential equation verified by \mathcal{V} is of order 1:

```
> ggDiffEq:= listtodiffeq(ggSeq, y(x));
```

$$ggDiffEq := \left[\left\{ y(0) = 1, -1 + 18x + (1 - 18x + 8x^2)y(x) + (12x^2 - 4x^3 - x) \left(\frac{\partial}{\partial x} y(x) \right) \right\}, ogf \right]$$

From this equation and the condition $\text{coeff}(y(x), x, 1) = 1$, we get the following closed form:

```
> closedForm:= dsolve(ggDiffEq[1], y(x));
subs(_C1=-1/2, closedForm);
```

$$closedForm := y(x) = 1 + \frac{3}{2}x - x^2 + x \sqrt{-12x + 4x^2 + 1} _C1$$

$$y(x) = 1 + \frac{3}{2}x - x^2 - \frac{1}{2}x \sqrt{-12x + 4x^2 + 1}$$

As shown in [FlaNo97], this corresponds to the general term:


```
> cn:=proc(n)
local k,l;
sum((-1)^k*product(2*l-1,l=1..n-k-1)/(factorial(k)*factorial(n-2*k))*3^(n-2*k)*2^(-k-2),
k=0..iquo(n,2))
end;
```

```
> gn:=proc(n) 2^n*cn(n-1) end;
```

```
gn := proc(n) 2^n*cn(n-1) end
```

```
> seq(gn(i), i=3..10);
```

```
8, 48, 352, 2880, 25216, 231168, 2190848, 21292032
```

Dissections and partitions

Dissections

A dissection of a convex polygon $P_n = \{v_1, \dots, v_n\}$ is a partition of the polygon into polygonal regions by means of non-crossing diagonals. If the polygonal region containing the edge $v_1 v_2$ has $r+1$ sides, one gets a bigger dissection by replacing the r edges by a dissection. So that a dissection is either an edge connecting two vertices or a sequence of dissections. The tricky point in sequencing 2 dissections consists in not counting the same vertex twice, as for the connected graphs above. When sequencing dissections, we therefore assume that each dissection provides its rightmost point while the leftmost one is the rightmost point of the dissection to the left in the sequence. With this convention, the number of dissections of size i actually counts the number of dissections of a polygon with $i+1$ vertices, and the grammar is:

```
> dissG:={Di=Union(Z, Sequence(Di, card >= 2))};
```

```
dissG := {Di = Union(Z, Sequence(Di, 2 ≤ card))}
```

The corresponding sequence, M2898 in [Sloa95] and related to Schroeder's second problem is:

```
> seq(count([Di, dissG], size=i), i=1..10);
```

```
1, 1, 3, 11, 45, 197, 903, 4279, 20793, 103049
```

Partitions

A non-crossing partition of size n is a partition of $[n]=\{1,2,\dots,n\}$ such that if $a < b < c < d$ and a block contains a and c , then no block contains b and d . Given a partition block, possibly empty, one gets a bigger partition by substituting to each vertex the product between Z and another partition. So that:

```
> partG:={P=Sequence(V), V = Prod(Z,P)};
```

```
partG := {P = Sequence(V), V = Prod(Z, P)}
```

We get the sequence of Catalan numbers, which can be checked in terms of generating functions:

```
> seq(count([P, partG], size=i), i=1..10);
```

```
1, 2, 5, 14, 42, 132, 429, 1430, 4862, 16796
```

> **gfsolve(partG, unlabelled, z);**

$$\{Z(z) = z, V(z) = \frac{1}{2} - \frac{1}{2}\sqrt{1-4z}, P(z) = \frac{1}{2} \frac{1 - \sqrt{1-4z}}{z}\}$$

Univariate asymptotics

Asymptotic counting: algorithm

Outline

As shown in [FlaNo97] for the six NC configurations we are interested in, the generating function $y(z)$ satisfies an algebraic equation. In the case of NC forests for example, we have:

> **eq:=y^3+(-z+z^2-3)*y^2+(z+3)*y-1;**

$$eq := y^3 + (-z + z^2 - 3)y^2 + (z + 3)y - 1$$

Like in many implicitly defined functions [Dr97, HaPa73], we expect *a priori* $y(z)$ to have locally an expansion of the square-root type, that is:

$$y(z) = c_0 + c_1 \sqrt{1 - \frac{z}{\rho}} + c_2 \left(1 - \frac{z}{\rho}\right) + O\left(\left(1 - \frac{z}{\rho}\right)^{\left(\frac{3}{2}\right)}\right)$$

By a singularity analysis at the dominant singularity ρ and denoting $\gamma = -\frac{c_1}{2}$, we get:

$$[z^n] y(z) = \frac{c_1 \rho^{(-n)} \left(1 + O\left(\frac{1}{n}\right)\right)}{\Gamma\left(-\frac{1}{2}\right) \sqrt{n^3}} \quad \text{or} \quad y_n = \frac{\gamma \rho^{(-n)} \left(1 + O\left(\frac{1}{n}\right)\right)}{\sqrt{\pi n^3}}$$

We now exemplify this method for the class of NC-forests. The singularities sought may arise at those points z such that $E(z,y)=0$ and $\frac{\partial}{\partial y} E(z,y) = 0$. In order to get the candidates z satisfying these conditions, we just have to eliminate y between the two previous equations through a resultant computation:

> **res_y:=resultant(eq,diff(eq,y),y);**
res_y:=expand(normal(res_y/gcd(res_y,diff(res_y,z))));
Omega[0]:=normal(res_y/z^ldegree(res_y));

$$res_y := -4z^3 + 32z^4 + 8z^5 - 5z^6$$

$$res_y := -4z + 32z^2 + 8z^3 - 5z^4$$

$$\Omega_0 := -4 + 32z + 8z^2 - 5z^3$$

And the singularities sought are the solutions of the previous equation whose modulus is smaller than 1. If the set found has cardinality one, we are done. It actually turns out that this is the case for all our configurations but the connected graphs where a 'ghost' singularity has to be eliminated by an external argument --see [FlaNo97]. In our case:

```
> rhonums:=[fsolve(Omega[0],z,complex)];
Omega[1]:=map(proc(x) if abs(x)<1 then x fi end,rhonums);
```

```
rhonums := [-1.930283307, .1215851069, 3.408698200]
```

```
Ω1 := [.1215851069]
```

And a convenient way to represent our singularity both in symbolic and numerical form is:

```
> rho_symb:=RootOf(Omega[0], z, op(Omega[1]));
```

```
rho_symb := RootOf(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)
```

Since in our case the dominant coefficient of the equation in y does not vanish, the function remains finite at its singularity. Its value $\tau = \lim_{z \rightarrow \rho} y(z)$ is also the quantity c_0 defined above.

Since the point $P(\rho, \tau)$ is a singularity, we have $E(\rho, \tau) = 0$ and $\left(\frac{\partial}{\partial y} E(z, y) \right)_{\rho, \tau} = 0$. Candidate

values for τ are therefore obtained as follows:

```
> deq:=subs(z=rho_symb, diff(eq,y));
print(`Candidate values for function at singularity`);
tau_vals:=[fsolve(deq,y)];
print(tau_vals);
```

```
deq := 3y2
+ 2 ( -RootOf(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + RootOf(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)2 - 3 ) y
+ RootOf(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + 3
```

Candidate values for function at singularity

```
tau_vals := [.8568817020, 1.214319744]
```

```
[.8568817020, 1.214319744]
```

Plugging back these candidates into the equation eq, we could get the correct one from a carefully controlled numerical analysis:

```
> eTau:=subs(z=rho_symb, eq);
[seq( evalf(subs(y=tau_vals[i], eTau)), i=1..nops(tau_vals))];
```

```
eTau := y3
```

$$\begin{aligned}
& + \left(-\text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 - 3 \right) y^2 \\
& + (\text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + 3) y - 1 \\
& \quad \quad \quad [.02283349037, -.69 \cdot 10^{-9}]
\end{aligned}$$

To get the constant c_1 we are missing, let us just compute the Puiseux expansions verified by $y(z)$ with the `algeqtoseries` procedure from the `gfun` package:

> **all_puis:=algeqtoseries(subs(z=rho_symb*(1-t^2),eq),t,y,6);**

$$\begin{aligned}
\text{all_puis} := & \left[\left(\frac{25}{37} + \frac{1}{37} \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - \frac{2}{37} \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 \right) \right. \\
& - \left(-25 - \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + 2 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 \right) \\
& \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \left(-12 - 49 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right. \\
& \left. - 4 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 + 4 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^3 \right) \\
& \left. \left(553 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - 432 \right) \right. \\
& - 1923 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 - 210 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^3 \\
& \left. + 136 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^4 \right) t^2 - 2 \\
& \left(-25 - \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + 2 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 \right) \\
& \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 \left(-62184756 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right. \\
& - 33723438 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \\
& - 152968 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \\
& - 1486900 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \\
& + 3704306 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \\
& + 18584788 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \\
& \left. - 21010686 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right)
\end{aligned}$$

$$\begin{aligned}
& -25667617 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \\
& + 43664 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \\
& - 81439279 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + 1181952 \Big) \Big/ \Big(\\
& 553 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - 432 \\
& - 1923 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - 210 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \\
& + 136 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \Big)^3 t^4 + O(t^6), \frac{43}{37} \\
& + \frac{18}{37} \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - \frac{35}{74} \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + \\
& \operatorname{RootOf} \left(1369_Z^2 + 5290 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - 228 \right. \\
& \left. + 981 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right) t + \left(\right. \\
& \frac{1365}{1369} \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - \frac{98}{1369} \\
& \left. - \frac{664}{1369} \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right) t^2 + \left(\frac{4657}{50653} \operatorname{RootOf} \left(1369_Z^2 \right. \right. \\
& \left. \left. + 5290 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - 228 \right. \right. \\
& \left. \left. + 981 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right) \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - \frac{5170}{50653} \right. \\
& \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \operatorname{RootOf} \left(1369_Z^2 \right. \\
& \left. + 5290 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - 228 \right. \\
& \left. + 981 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right) - \frac{24841}{50653} \operatorname{RootOf} \left(1369_Z^2 \right. \\
& \left. + 5290 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - 228 \right. \\
& \left. + 981 \operatorname{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right) \Big) t^3 + O(t^{7/2}) \Big)
\end{aligned}$$

and select those with a t^2 term --which corresponds to the square-root sought due to the change of variable $z = \rho(1 - t^2)$:

```

> #--convert into series i.e. remove the O()
all_puis_s:=map(eval, map2(subs, O=0, all_puis));
#--collect the coeff in t and select the non-null one(s)
c1:=map(proc(x) if x<>0 then x fi end, map(coeff, all_puis_s, t, 1));
print(`Constant`, evalf(c1,10));

```

$$\begin{aligned}
 \text{all_puis_s} := & \left[\left(\frac{25}{37} + \frac{1}{37} \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - \frac{2}{37} \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right)^2 \right. \\
 & - \left(-25 - \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + 2 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right)^2 \\
 & \left. \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \left(-12 - 49 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right. \right. \\
 & \left. \left. - 4 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + 4 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right)^3 \right) \\
 & / \left(553 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) - 432 \right. \\
 & - 1923 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 - 210 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^3 \\
 & \left. + 136 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^4 \right) t^2 - 2 \\
 & \left(-25 - \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) + 2 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right)^2 \\
 & \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^2 \left(-62184756 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069) \right)^2 \\
 & - 33723438 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^9 \\
 & - 152968 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^8 \\
 & - 1486900 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^7 \\
 & + 3704306 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^6 \\
 & + 18584788 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^5 \\
 & - 21010686 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^3 \\
 & - 25667617 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^10 \\
 & + 43664 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^4 \\
 & \left. - 81439279 \text{RootOf}(4 - 32_Z - 8_Z^2 + 5_Z^3, .1215851069)^4 + 1181952 \right) / \left(\right.
 \end{aligned}$$

$$\begin{aligned}
& 553 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - 432 \\
& - 1923 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - 210 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \\
& + 136 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \Big] t^4, \frac{43}{37} \\
& + \frac{18}{37} \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - \frac{35}{74} \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) + \\
& \operatorname{RootOf} \left(1369_z^2 + 5290 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - 228 \right. \\
& \left. + 981 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \right) t + \left(\right. \\
& \frac{1365}{1369} \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - \frac{98}{1369} \\
& \left. - \frac{664}{1369} \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \right) t^2 + \left(\frac{4657}{50653} \operatorname{RootOf} \left(1369_z^2 \right. \right. \\
& \left. \left. + 5290 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - 228 \right. \right. \\
& \left. \left. + 981 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \right) \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - \frac{5170}{50653} \right. \\
& \left. \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \operatorname{RootOf} \left(1369_z^2 \right. \right. \\
& \left. \left. + 5290 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - 228 \right. \right. \\
& \left. \left. + 981 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \right) - \frac{24841}{50653} \operatorname{RootOf} \left(1369_z^2 \right. \right. \\
& \left. \left. + 5290 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - 228 \right. \right. \\
& \left. \left. + 981 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \right) \right) t^3 \Big] \\
c1 := & \left[\operatorname{RootOf} \left(1369_z^2 + 5290 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) - 228 \right. \right. \\
& \left. \left. + 981 \operatorname{RootOf}(4 - 32_z - 8_z^2 + 5_z^3, .1215851069) \right) \right] \\
& \text{Constant, } [-.1493185566]
\end{aligned}$$


Should we have found several expansions containing a t term, the right one could have been

selected from the formula $c_1 = -\sqrt{\frac{2 p_{1,0}}{p_{0,2}}}$ and $p_{i,j} = \frac{\operatorname{Diff}(eq, z^i y^j)}{i! j!}$. Also note that the

minus sign has to be adopted for c_1 since the generating function increases with its argument.

We can locally plot the algebraic curve in the neighborhood of the singularity:

```
> implicitplot(eq, z=-.1..0.15, y=-1..1.3, numpoints=5000);
```



Putting everything together, we get the following procedure which we apply to the remaining NC configurations.

Implementation

We first look for the singularity(ies) of smallest modulus(i):

```
> locateDominantSing := proc(eq, y, z)
local res_y, Omega, rhonums,
rho_symb;

#--we look for the z such that eq and diff(eq,y) have a common root y
res_y:=resultant(eq,diff(eq,y),y);
res_y:=expand(normal(res_y/gcd(res_y,diff(res_y,z))));
Omega[0]:=normal(res_y/z^ldegree(res_y));
rhonums:=[fsolve(Omega[0],z,complex)];

#--we are just interested in roots of modulus < 1
Omega[1]:=map(proc(x) if abs(x)<1 then x fi end,rhonums);
if nops(Omega[1])<>1 then
ERROR( `More than one root of modulus < 1: not implemented!` ) fi;

rho_symb:=RootOf(Omega[0], z, op(Omega[1]));
print( `Singularity is` ); print(rho_symb);

rho_symb
end;
```

And then the asymptotic expansion. The values returned are:

- 1.the symbolic expressions for the c_1 constant
- 2.the dominant singularity ρ
- 3.the Puiseux expansion

```
> singExpansion:=proc(eq, y, z)
local rho_symb,
deq, tau_vals,
all_puis, all_puis_s, c1;
```



```

##--numerical and symbolic representation of the singularities
rho_symb:=locateDominantSing(eq, y, z);

##--values of the function at the singularity
deq:=subs(z=rho_symb, diff(eq,y));
print(`Candidate values for function at singularity`);
tau_vals:= [fsolve(deq,y)];
print(tau_vals);

##--Puiseux expansion; we expect a single exp. to have a t term
all_puis:=algeqtoseries(subs(z=rho_symb*(1-t^2),eq),t,y,6);

##--convert into series i.e. remove the O()
all_puis_s:=map(eval, map2(subs, O=0, all_puis));
##--collect the coeff in t and select the non-null one(s)
c1:=map(proc(x) if x<>0 then x fi end, map(coeff, all_puis_s, t, 1));

if nops(c1)<>1 then
print(`Problem in singular expansion`); print(all_puis); RETURN(FAIL)
else
c1:=op(c1); if evalf(c1)>0 then c1:=-c1 fi
fi;

##--returns the constant c1 and the singularity in symbolic forms
print(`c1 constant`, c1,evalf(c1,20));
[c1,rho_symb, my_puis]
end;

```

Applications to non-crossing configurations

We now present the whole chain that goes from the grammars specification to the asymptotic machinery. More precisely, for each of the NC configurations studied above, we:

- 1.call `Combstruct[gfeqns]` to retrieve the system of equations verified by the generating functions associated with the grammar,
- 2.compute the algebraic equation verified by a given generating function with `gfun[algfuntoalgeq]`,
- 3.feed this equation to the asymptotic machinery developed above.

Trees

Generating functions associated with the grammar:

> **gfeqns(tbf, unlabelled, z);**

$$\left[B(z) = F(z)^2 Z(z), F(z) = \frac{1}{1 - B(z)}, Z(z) = z, T(z) = \frac{Z(z)}{1 - B(z)} \right]$$

> **treesSys:=gfsolve(tbf, unlabelled, z);**

$$treesSys := \left[F(z) = \text{RootOf}(-Z + Z^3 z + 1), T(z) = -\frac{z}{-1 + \text{RootOf}(-Z + Z^3 z + 1) z} \right]$$

$$B(z) = \text{RootOf}(-z + z^3 z + 1) \quad z, Z(z) = z \quad \left. \vphantom{B(z)} \right\}$$

Algebraic equation:

> **trees:=algfuntoalgeq(subs(treesSys,T(z)), y(z));**

$$\text{trees} := -y z + z^2 + y^3$$

Asymptotic machinery:

> **resTrees:=singExpansion(trees, y, z):**

Singularity is

$$\frac{4}{27}$$


Candidate values for function at singularity

$$[-.2222222222, .2222222222]$$

$$\text{cl constant, RootOf}(-4 + 243 z^2), -.12830005981991683656$$

Plot of the singularity:

> **implicitplot(trees, z=-0.2..0.2, y=-1.1, numpoints=5000);**



Forests

> **forests:=y^3+(-z+z^2-3)*y^2+(z+3)*y-1;**

$$\text{forests} := y^3 + (-z + z^2 - 3)y^2 + (z + 3)y - 1$$

> **resForests:=singExpansion(forests, y, z):**

Singularity is

$$\text{RootOf}(4 - 32 z - 8 z^2 + 5 z^3, .1215851069)$$

Candidate values for function at singularity

[.8568817020, 1.214319744]

$$c1 \text{ constant, RootOf}\left(1369 _Z^2 + 5290 \text{RootOf}(4 - 32 _Z - 8 _Z^2 + 5 _Z^3, .1215851069) - 228\right. \\ \left.+ 981 \text{RootOf}(4 - 32 _Z - 8 _Z^2 + 5 _Z^3, .1215851069)\right), -1.4931855684380904697$$

Connected graphs

Generating functions associated with the grammar:

> **CGSys:=gfsolve(ar, unlabelled, z);**

$$CGSys := \left\{ \begin{array}{l} Z(z) = z, EA(z) = \text{RootOf}(_Z - 3 _Z^2 + 2 _Z^3 - z), \\ C(z) = -\frac{z \left(1 - 3 \text{RootOf}(_Z - 3 _Z^2 + 2 _Z^3 - z) + 2 \text{RootOf}(_Z - 3 _Z^2 + 2 _Z^3 - z) \right)^2}{-1 + 3 \text{RootOf}(_Z - 3 _Z^2 + 2 _Z^3 - z) - 2 \text{RootOf}(_Z - 3 _Z^2 + 2 _Z^3 - z) + z} \end{array} \right.$$

Algebraic equation:

> **CG:=algfuntoalgeq(subs(CGSys,C(z)), y(z));**

$$CG := y^3 + y^2 - 3yz + 2z^2$$

Asymptotic machinery:

> **resCG:=singExpansion(CG, y, z);**

Error, (in locateDominantSing) More than one root of modulus < 1: not implemented!

As observed in [FlaNo97], due to the two singularities found, we need additional information to select the right one. Eliminating the negative value and performing the previous computations yields the following c_1 and ρ :

> **resCG[1]:=-RootOf(54*_Z^2-7+72*RootOf(-1+108*_Z^2,.96225044864937627418e-1));**


$$resCG_1 := -\text{RootOf}(54 _Z^2 - 7 + 72 \text{RootOf}(-1 + 108 _Z^2, .096225044864937627418))$$

> **resCG[2]:=RootOf(-1+108*_Z^2,.96225044864937627418e-1);**

$$resCG_2 := \text{RootOf}(-1 + 108 _Z^2, .096225044864937627418)$$

Plot of the singularity:

> **implicitplot(CG, z=-0.05..0.15, y=-.05..0.2, numpoints=10000);**



General graphs

Generating functions associated with the grammar:

> **GGSys:=gfsolve(br, unlabelled, z);**

$$GGSys := \{EA(z) = \frac{1}{4} - \frac{1}{2}z - \frac{1}{4}\sqrt{1 - 12z + 4z^2}, V(z) = z \left(1 + z + 2 \left(\frac{1}{4} - \frac{1}{2}z - \frac{1}{4}\sqrt{1 - 12z + 4z^2} \right) z \right),$$

$$G(z) = 1 + z + 2 \left(\frac{1}{4} - \frac{1}{2}z - \frac{1}{4}\sqrt{1 - 12z + 4z^2} \right) z, Z(z) = z\}$$

Algebraic equation:

> **GG:=algfuntoalgeq(subs(GGSys,G(z)), y(z));**

$$GG := y^2 + (-2 - 3z + 2z^2)y + 1 + 3z$$

Asymptotic machinery:

> **resGG:=singExpansion(GG, y, z):**

Singularity is

$$\text{RootOf}(1 - 12_Z + 4_Z^2, .08578643763)$$


Candidate values for function at singularity


$$[1.121320344]$$

$$c1 \text{ constant, RootOf}(8_Z^2 + 198 \text{RootOf}(1 - 12_Z + 4_Z^2, .08578643763) - 17), -.042257173759048982037$$

Plot of the singularity:

> **implicitplot(GG, z=-0.2..0.1, y=0.9..1.2, numpoints=10000);**





Dissections

> **dissSys:=gfsolve(dissG, unlabelled, z);**

$$dissSys := \{Di(z) = \frac{1}{4} + \frac{1}{4}z - \frac{1}{4}\sqrt{1 - 6z + z^2}, Z(z) = z\}$$

> **diss:=algfuntoalgeq(subs(dissSys,Di(z)), y(z));**

$$diss := 2y^2 + (-1 - z)y + z$$

> **resDiss:=singExpansion(diss, y, z):**

Singularity is

$$\text{RootOf}(1 - 6_Z + _Z^2, .1715728753)$$

Candidate values for function at singularity

$$[.2928932188]$$

$$c1 \text{ constant}, \text{RootOf}(8_Z^2 + 3 \text{RootOf}(1 - 6_Z + _Z^2, .1715728753) - 1), -.24629285775235400967$$

It should be noticed that the equation obtained in the paper and stated below is slightly different. But remember that z^i counts here the number of dissections of a polygon with $i + 1$ vertices. Of course, we still have the same singularity:

> **diss2:=2*y^2-z*(1+z)*y+z^3:**

> **resDiss2:=singExpansion(diss2, y, z):**

Singularity is

$$\text{RootOf}(1 - 6_Z + _Z^2, .1715728753)$$


Candidate values for function at singularity

$$[.05025253170]$$

$$c1 \text{ constant}, \text{RootOf}(8_Z^2 + 99 \text{RootOf}(1 - 6_Z + _Z^2, .1715728753) - 17), -.042257173759048980558$$

Plot of the singularity:

> **implicitplot(diss, z=-1..1, y=-1..1, numpoints=1000);**



Partitions

> **partsSys:=gfsolve(partG, unlabelled, z);**

$$\text{partsSys} := \{Z(z) = z, V(z) = \frac{1}{2} - \frac{1}{2}\sqrt{1-4z}, P(z) = \frac{1}{2} \frac{1 - \sqrt{1-4z}}{z}\}$$

> **parts:=algfuntoalgeq(subs(partsSys,P(z)), y(z));**

$$\text{parts} := 1 + y^2 z - y$$

> **resParts:=singExpansion(parts, y, z);**

Singularity is

$$\frac{1}{4}$$

Candidate values for function at singularity

[2.]

Problem in singular expansion

$$[2 - 2t + 2t^2 - 2t^3 + O(t^{7/2}), 2 + 2t + 2t^2 + 2t^3 + O(t^{7/2})]$$


resParts := FAIL


Here we are in trouble since there are 2 Puiseux expansions with t terms. But only the first one makes sense since the generating function increases with its argument. So that:

> **resParts[1]:=2: resParts[2]:=1/4:**

Plot of the singularity:

> **implicitplot(parts, z=-2..1, y=-5..5, numpoints=1000);**





Exact and estimated values: comparisons

We have already seen that an estimate of the number of objects of size n is given by:

```
> nbEnt:=proc(c1,rho, n)
evalf(c1*rho^(-n)/(2*sqrt(Pi*n^3)))
end;
```

```
nbEnt := proc(c1, rho, n) evalf(1 / 2*c1*rho^(-n) / sqrt(pi*n^3)) end
```

Comparisons with the exact values are as follows. It is interesting to observe that these values are within about 2% in any case:

Trees

```
> exactTr:=evalf(count([T,tbf], size=100));
```

```
exactTr := .31112294248977312479 1079
```

```
> estimTr:=nbEnt(resTrees[1],resTrees[2],100);
```

```
estimTr := .30831823414949327377 1079
```

```
> exactTr/estimTr;
```

```
1.0090967968469874574
```

Forests

```
> exactFo:=evalf(count([F,fo], size=100));
```

```
exactFo := .13843759166925326460 1088
```

```
> estimFo:=nbEnt(resForests[1], resForests[2], 100);
```

```
estimFo := .13692122170039683439 1088
```

```
> exactFo/estimFo;
```

```
1.0110747621882491234
```

Connected graphs

```
> exactCo:=evalf(count([C,ar],size=100));
```

```
exactCo := .48999935708003238530 1097
```

> **estimCo:=nbEnt(resCG[1],resCG[2],100);**

estimCo := .48243456359851996544 10⁹⁷

> **exactCo/estimCo;**

1.0156804550343283644

General graphs

> **exactGG:=evalf(count([G, br], size=100));**

exactGG := .55274982765861142397 10¹⁰²

> **estimGG:=nbEnt(resGG[1], resGG[2], 100);**

estimGG := .54254048238981239890 10¹⁰²

> **exactGG/estimGG;**

1.0188176654096452581

Dissections

> **exactDiss:=evalf(count([Di, dissG], size=100));**

exactDiss := .25033275556682302012 10⁷³

> **estimDiss:=nbEnt(resDiss[1], resDiss[2], 100);**

estimDiss := .24945025571433850936 10⁷³

> **exactDiss/estimDiss;**

1.0035377789048856216

Partitions

> **exactPart:=evalf(count([P, partG], size=100));**

exactPart := .89651994709013149669 10⁵⁷

> **estimPart:=nbEnt(resParts[1], resParts[2], 100);**

estimPart := .90661770597752568402 10⁵⁷

> **exactPart/estimPart;**

.98886216448143744662

>

References

- [Dr97] M. Drmota, Systems of functional equations, *Random Structures and Algorithms* 10, 1-2, 1997.
- [FlaNo97] P. Flajolet and M. Noy, Analytic Combinatorics of Non-Crossing Configurations, *Rapport de Recherche INRIA No. 3196*, 1997.
- [HaPa73] F. Harary and E.M. Palmer, *Graphical Enumeration*, Academic Press, 1973.
- [Sloa95] N.J.A. Sloane and S. Plouffe, *The Encyclopedia of Integer Sequences*, Academic Press, 1995.